Lab #5
TU850/1 - Creative Coding
You are welcome to submit this lab before Monday at 12 noon (and there’s no need to come into the lab), or during the lab before 2pm.
QUESTION 1
	· Piet Mondrian (7th March 1872 – 1st February 1944) was a Dutch painter and art academic who is one of the greatest artists of the 20th century. 
· He was one of the pioneers of 20th-century abstract art, as he changed his artistic direction from figurative painting to an increasingly abstract style, until he reached a point where his artistic vocabulary was reduced to simple geometric elements.
· He evolved a non-representational form which was called the new 'pure plastic art' which he believed was necessary in order to create 'universal beauty'.
	[image: Piet Mondrian - Wikipedia]


To express this “universal beauty”, Mondrian eventually decided to limit his formal vocabulary to the three primary colours (red, blue, and yellow), the three primary values (black, white, and grey), and the two primary directions (horizontal and vertical).
	Early work
	
	Later work

	[image: Piet Mondrian painting Spring Sun (Lentezon): Castle Ruin: Brederode in the Dallas Museum of Art]

	
	[image: Piet Mondrian abstract painting Tableau I, from 1921]


	[image: Piet Mondrian painting Evening; Red Tree in the Gemeentemuseum Den Haag]
	
	[image: Piet Mondrian abstract painting Composition II in Red, Blue, and Yellow, 1930]



The Processing 4 code below generates paintings in the style of Mondrian, so for this question, I want you to do two things:
1. Run the code three (3) times, and screengrab each one. 
2. Now, change the following line of code: float minHeight = height / 4.0; 
from 4.0 to both 2.0 and to 6.0, and run each one of those twice, and screengrab.
	[image: Processing 4]
	Processing 4

	void setup() {
  size(600, 600);
  noLoop();  // Draw only once
}
void draw() {
  background(255);  // White background
  float minHeight = height / 4.0;  // Ensuring at least 4 rows
  float y = 0;
  
  for (int i = 0; i < 4; i++) {
    float h = random(minHeight * 0.8, minHeight * 1.2);  // Slight variation in row height
    if (i == 3) h = height - y;  // Ensure last row fits perfectly
    drawRow(0, y, width, h);
    y += h;
  }
}
void drawRow(float x, float y, float w, float h) {
  if (w > 80 && random(1) > 0.3) {  
    // Split into multiple columns within the row
    float startX = x;
    while (startX + 80 < x + w) {  // Ensure minimum width of 80
      float rectW = random(80, (x + w - startX) / 2);
      if (startX + rectW > x + w - 80) rectW = x + w - startX; // Fit last column
      drawRectangle(startX, y, rectW, h);
      startX += rectW;
    }
  } else {
    drawRectangle(x, y, w, h);
  }
}
void drawRectangle(float x, float y, float w, float h) {
  stroke(0);
  strokeWeight(8);
  fill(getMondrianColor());
  rect(x, y, w, h);
}
color getMondrianColor() {
  color[] palette = {color(255, 0, 0), color(0, 0, 255), color(255, 255, 0), color(255)};
  return palette[int(random(palette.length))];
}



QUESTION 2
	· Georges Seurat (2nd December 1859 – 29th March 1891) was a French painter and is considered one of the greatest artists of the 20th century.
· He devised several painting techniques, including pointillism which is a technique that relies on the ability of the eye and mind of the viewer to blend the colour spots into a fuller range of tones. 
· His art combines qualities that are usually thought of as opposed and incompatible: on the one hand, his extreme and delicate sensibility, on the other, a passion for logical abstraction and an almost mathematical precision of mind.
	[image: Georges Seurat]



His large-scale work “A Sunday Afternoon on the Island of La Grande Jatte” (1884–1886) changed the direction of modern art, and is one of the icons of late 19th-century painting.


[image: undefined]




[image: A person holding a bouquet of flowers

AI-generated content may be incorrect.]


[image: The Circus Parade by Georges Seurat]
[image: A close-up of a painting

AI-generated content may be incorrect.]

	Let’s try to draw one of the trees from “A Sunday Afternoon”, pictured here, using this point-by-point technique.

· So, here’s the code in Processing 4, run it once, and do a screengrab of it.

· Now change the 0.55 on the fourth line below, first to 0.45 and then to 0.65, and screengrab those.

	[image: ]

	[image: Processing 4]
	Processing 4

	void setup() {
  size(400, 400);
  background(255);
  drawPointillistTree(width * 0.5, height * 0.55);
}
void drawPointillistTree(float x, float y) {
  int numDots = 24000; // doubled for density
  float trunkWidth = 50;   // doubled
  float trunkHeight = 120; // doubled
  float foliageRadius = 140; // doubled

  // Draw trunk with texture
  for (int i = 0; i < numDots / 4; i++) {
    float px = x + random(-trunkWidth / 2, trunkWidth / 2);
    float py = y + random(0, trunkHeight);
    stroke(120 + random(-30, 30), 60 + random(-20, 20), 20);
    point(px, py);
  }
  // Draw foliage above the trunk
  for (int i = 0; i < numDots; i++) {
    float angle = random(TWO_PI);
    float radius = sqrt(random(0, 1)) * foliageRadius;
    float px = x + radius * cos(angle);
    float py = y - trunkHeight + radius * sin(angle);

    stroke(getSeuratTreeColor(px, py));
    point(px, py);
  }
}
color getSeuratTreeColor(float x, float y) {
  float noiseValue = noise(x * 0.02, y * 0.02);
  if (noiseValue < 0.2) {
    return color(20, 100, 20); // Dark green shadows
  } else if (noiseValue < 0.5) {
    return color(34, 139, 34); // Rich green
  } else if (noiseValue < 0.8) {
    return color(50, 205, 50); // Medium green
  } else {
    return color(173, 255, 47); // Bright highlights
  }
}


QUESTION 3
	· Pablo Picasso (25th October 1881 – 8th April 1973) was a Spanish painter, sculptor, printmaker, ceramicist, and theatre designer who spent most of his adult life in France. 
· One of the most influential artists of the 20th century, he is known for co-founding the Cubist movement, the invention of constructed sculpture, the co-invention of collage, and for the wide variety of styles that he helped develop and explore. 
· Among his most famous works is the anti-war painting Guernica (1937), a dramatic portrayal of the bombing of Guernica by German and Italian air forces during the Spanish Civil War.
	[image: ]


Picasso didn’t think of his work as evolving linearly, instead he use whichever technique suited the subject of his painting: "The several manners I have used in my art must not be considered as an evolution, or as steps towards an unknown ideal of painting ... If the subjects I have wanted to express have suggested different ways of expression I have never hesitated to adopt them."
	[image: Don Quixote (Picasso) - Wikipedia]
	[image: Pablo Picasso's Child with a Dove to ...]
	[image: Self Portrait, 1901 by Picasso]



Guernica
[image: The Beauty of Picasso's Guernica | Toward Beauty]
The Processing 4 code below generates images in the style of Guernica, I just want you to run this program twice, and screengrab the outputs (they will be very similar to each other).
	[image: Processing 4]
	Processing 4

	void setup() {
  size(1200, 500);
  background(255);
  noLoop(); // Only draw once
}

void draw() {
  drawGuernicaAbstract();
}

void drawGuernicaAbstract() {
  noStroke();

  // Background sections with random brightness variations
  fill(random(30, 70)); rect(0, 0, width/3, height);
  fill(random(80, 120)); rect(width/3, 0, width/3, height);
  fill(random(150, 220)); rect(2 * width/3, 0, width/3, height);

  // Abstract face (randomized position)
  fill(0);
  triangle(random(250, 350), random(50, 150), 
           random(300, 400), random(250, 350), 
           random(180, 250), random(200, 300));

  // Abstract body with slight randomness
  triangle(random(480, 520), random(80, 120), 
           random(620, 680), random(250, 350), 
           random(480, 520), random(350, 450));

  // White contrast shapes
  fill(255);
  ellipse(random(700, 800), random(150, 250), 100, 150); // Eye or face
  ellipse(random(850, 950), random(250, 350), 80, 120);  // Another figure
  
  // Abstract details
  fill(0);
  rect(random(580, 620), random(220, 280), 120, 60); 
  rect(random(780, 820), random(380, 420), 50, 100);

  // Jagged fragmented shape with random points
  fill(100);
  beginShape();
  for (int i = 0; i < 5; i++) {
    vertex(random(50, 300), random(300, 450));
  }
  endShape(CLOSE);

  // Lightbulb motif with randomized size
  fill(255);
  float bulbSize = random(50, 80);
  ellipse(950, 80, bulbSize, bulbSize);
  fill(0);
  triangle(940, 60, 960, 60, 950, 40);

  // Stylized "screaming mouth" with randomized teeth
  fill(255);
  arc(500, 350, 80, 40, 0, PI);
  fill(0);
  for (int i = 0; i < 4; i++) {
    rect(470 + (i * 15), 340, random(5, 10), 5); // Random width for teeth
  }

  // Random jagged chaos lines
  stroke(0);
  strokeWeight(3);
  for (int i = 0; i < 5; i++) {
    line(random(100, 900), random(50, 450), random(100, 900), random(50, 450));
  }
}






Submission:
Submit the screengrabs, and paste it all onto your Template document.
	e-mail me a completed solution to each of the above programs in your Template document. The Template document should be renamed as follows:
· Surname_Firstname_Student#__Lab5.pdf
· for example: Smith_John_D1234567_Lab5.pdf
Send it to Damian.X.Gordon@tudublin.ie with subject heading “DT850 CC Lab #5”, and put it in Brightspace as well.



You should be able to fit it in all two pages, like this:

[image: A book with pictures of different colors and shapes

AI-generated content may be incorrect.]

"Study hard what interests you the most in the most undisciplined, irreverent and original manner possible." - Richard P. Feynman

image3.jpeg




image4.jpeg




image5.jpeg




image6.png




image7.jpeg




image8.jpeg




image9.png




image10.jpeg




image11.png




image12.png




image13.png




image14.jpeg




image15.jpeg




image16.jpeg




image17.jpeg




image18.png




image1.jpeg




image2.jpeg




